
Введение
В языке программирования Паскаль  приходится работать с различными данными, т.е константами, переменными, значениями функций или выражений. Все они характеризуются своими типами. Тип определяет множество допустимых значений, которые может иметь тот или иной объект, а также множество допустимых операций,  которые применимы к нему. Кроме того, тип определяет также и формат внутреннего представления данных в памяти ПК.
Паскаль характеризуется разветвленной структурой типов данных.Типы данных делятся на простые, структурированные, указатели, строки, процедурные, объекты, варианты,  классы. 

Любой из структурированных типов (массивы, записи, множества, файлы) характеризуется множественностью образующих этот тип элементов.

Множества – это наборы однотипных, логически связанных друг с другом объектов; файл – именованная область внешней памяти ПК; массивы объединяют однородные единицы информации – элементы одного и того же типа. Но многообразие информации нельзя свести только к какому-то одному типу данных. Например, указывая положение точки в пространстве, мы можем воспользоваться одним и тем же типом для указания ее координат, но, описывая человека, мы должны указать его имя, рост, цвет глаз и волос, то есть в одном описании объединим разнородную информацию. Точно так же, описывая автомобиль, мы укажем не только его марку, но и год выпуска, модификацию, да и цвет кузова может нас заинтересовать. Составляя автоматизированный каталог книгохранилища, мы для каждой книги должны указать ее название, имя автора, область знания, количество страниц, год издания, а также, возможно, признак нахождения на руках или в хранилище.

Данные такого рода, описывающие существенные стороны того или иного объекта путем включения в описание нескольких, часто разнотипных, элементов, называют записью (record). В языке Паскаль запись определяется путем указания служебного слова record и перечисления входящих в запись элементов с указанием типов этих элементов. 
Запись Паскаля

Запись Паскаля – структурированный комбинированный тип данных, состоящий из фиксированного числа компонент (полей) разного типа.
Например, анкетные данные о студенте вуза могут быть представлены в виде информационной структуры 

Такая структура называется двухуровневым деревом. В Паскале эта информация может храниться в одной переменной типа record (запись). Задать тип можно следующим образом: 

type < имя _ типа >=record
   <имя_поля1>: тип; 
   <имя_поля2>: тип; 
   …………………. 
   <имя_поля K >: тип 
end ; 

где record – служебное слово, а <имя_типа> и <имя_поля> - правильные идентификаторы языка Паскаль. 

Описание анкеты студента в Паскале будет выглядеть так:

Пример фрагмента программы описания записи Паскаля 

Type anketa=record
   fio: string[45]; 
   pol: char; 
   dat_r: string[8]; 
   adres: string[50]; 
   curs: 1..5; 
   grupp: string[3];
end; 

Такая запись Паскаля, так же как и соответствующее ей дерево, называется двухуровневой. 

Поля записи Паскаля могут иметь любой тип, в частности сами могут быть записями. Такая возможность используется в том случае, когда требуется представить многоуровневое дерево (более 2 уровней). Например, те же сведения о студентах можно отобразить трехуровневым деревом. 

Такая организация данных позволит, например, делать выборки по году рождения или по городу, где живут студенты. В этом случае описание соответствующей записи в Паскале будет выглядеть так: 

Пример фрагмента программы описания записи Паскаля 

Type anketa1=record
   fio: string[45]; 
   pol: char; 
   dat_r: record;
   god: integer; 
   mes: string[10]; 
   den: 1..31;
end; 
adres: record
   gorod: string[25]; 
   ulica: string [20]; 
   dom, kv: integer;
end; 
curs: 1..5; 
   grupp: string[3]; 
end; 

Поля

После того, как определен тип записи Паскаля, можно определять переменную этого типа. Переменная определяется путем задания ее идентификатора и указания типа.

var 
student: anketa; 
student 1: anketa 1; 

Теперь нам нужно узнать, как правильно получать доступ к элементам записи Паскаля. Элементы записи называются полями, а обращение к ним производится через использование их имен – идентификаторов полей. Практически, поля записи обрабатываются точно так же, как и любые другие переменные. Но в отличие от обычной переменной имена полей должны предваряться ссылкой на идентификатор записи Паскаля и отделяться от него точкой. Такая запись называется уточняющий идентификатор: 

<имя_записи>.<имя_поля> 

Например, чтобы обратиться к полю curs переменной student , необходимо указать следующее составное имя:

student.curs :=3; 

Для того чтобы обратиться к полю god в записи student 1, необходимо записать уточняющий идентификатор, состоящий из трех имен: 

student1.dat_r.god:=1982; 

Использование полей записи Паскаля в выражениях и условиях идентично использованию обычных переменных. 

Операции над записями Паскаля 

Единственная операция, которую можно произвести над однотипными записями Паскаля – это присваивание.

Все другие операции производятся над отдельными полями записи. 

Рассмотрим для начала простейший пример заполнения записи Паскаля и вывода ее на экран.

Пусть нам необходимо заполнить сведения о студенте (Ф.И.О., дата рождения, адрес, курс и группа), а затем вывести эти сведения на экран. 

Пример программы c записью Паскаля

program primer1; 
type anketa=record 
   fio: string[45]; 
   dat_r: string[8]; 
   adres: string[50]; 
   curs: 1..5; 
   grupp: string[3]
end; 
var student: anketa; 
begin 
   writeln (‘введите сведения о студенте’); 
   {обратите внимание, ввод каждого поля осуществляется отдельно} 
   writeln (‘введите фамилию, имя и отчество’); 
   readln (student.fio); 
   writeln (‘введите дату рождения’); 
   readln (student.dat_r); 
   writeln (‘введите адрес’); 
   readln(student.adres); 
   writeln (‘введите курс’); 
   readln(student.curs); 
   writeln (‘введите группу’); 
   readln (student.grupp); 
   writeln (‘ввод закончен’); 
   writeln ; 
   {обратите внимание, что вывод записи осуществляется по полям} 
   writeln (‘фамилия студента: ’, student . fio ); 
   writeln(‘ дата рождения : ’, student.dat_r); 
   writeln(‘ адрес : ’, student.adres); 
   writeln(‘ курс : ’, student.curs); 
   writeln(‘ группа : ’, student.grupp); 
end. 

А теперь слегка усложним задачу. Пусть нам необходимо иметь сведения о многих студентах, например, нашего факультета. Следовательно, необходимо организовать массив записей Паскаля. А затем из общего списка вывести фамилии студентов 2-го курса. 

Пример программы c записью Паскаля 

program primer 2 ; 
type anketa=record 
   fio: string[45]; 
   dat_r: string[8]; 
   adres: string[50]; 
   curs: 1..5; 
   grupp: string[3]
end; 
var student: array [1..100] of anketa; 
   I: integer; 
begin 
   {последовательно вводим каждую запись} 
   for I:=1 to 100 do 
   begin 
      writeln (‘введите сведения о’, I , ‘-м студенте’); 
      writeln (‘введите фамилию, имя и отчество’); 
      readln (student[I].fio); 
      writeln (‘введите дату рождения’); 
      readln (student[I].dat_r); 
      writeln (‘введите адрес’); 
      readln(student[I].adres); 
      writeln (‘введите курс’); 
      readln(student[I].curs); 
      writeln (‘введите группу’); 
      readln (student[I].grupp); 
   end; 
   writeln (‘ввод закончен’); 
   writeln ; 
   {просматриваем массив записей и выбираем только студентов 2-го курса } 
   for I:=1 to 100 do 
      if student[I].curs=2 then 
         writeln(‘ фамилия студента : ’, student[I].fio); 
end. 

Префикс
Префикс – обязательная предшествующая часть составного идентификатора для имен полей в структуре типа запись Паскаля. Очень часто у программиста возникает желание не указывать префикс в имени полей, например, когда идет постоянное использование одних и тех же записей. В языке Паскаль предусмотрена такая возможность, реализуемая при помощи оператора присоединения, который в общем виде выглядит так:

with <имя_записи> do <действие с полем записи>; 

Следует обратить внимание на то, что после служебного слова do может стоять только один оператор, но он может быть составным (любая последовательность операторов, заключенная в операторные скобки begin end ). 

Например, фрагмент из предыдущей программы с использованием оператора присоединения будет выглядеть так: 

Пример фрагмента программы c записью и префиксом Паскаля 

for I:=1 to 100 do 
   with student[I] do 
   begin 
      writeln (‘введите сведения о’, I , ‘-м студенте’); 
      writeln (‘введите фамилию, имя и отчество’); 
      readln (fio); 
      writeln (‘введите дату рождения’); 
      readln (dat_r); 
      writeln (‘введите адрес’); 
      readln(adres); 
      writeln (‘введите курс’); 
      readln(curs); 
      writeln (‘введите группу’); 
      readln (grupp); 
   end; 

 Даты каких-либо событий можно описать следующим образом:

type date = record

    month: 1..12;

    day: 1..31;

    year: integer

end;

var d: date;

В этом примере описан тип date и переменная d, принадлежащая этому типу.

Переменная d описана как запись, состоящая из трех полей: month, day и year. Каждое поле содержит соответственно данные: целое число в пределах от 1 до 12 (номер месяца), целое число от 1 до 31 (число), целое число (год).

Поле day переменной d записывается как d.day.

Например, чтобы заслать в d дату 12.01.2003, надо выполнить следующие операторы:

d.month := 1;

d.day := 12;

d.year := 2003;

Пример. Вычислить сумму s двух комплексных чисел x = 2 + 7i и y = 6 + 3i (т.е. x, y, s: complex;).

x.re := 2.0; x.im := 7.0;

y.re := 6.0; y.im := 3.0;

s.re := x.re + y.re;

s.im := x.im + y.im;

Запись может быть компонентой других структур. Например, введем тип family (семья: отец, мать, 1-й ребенок, 2-й ребенок):

type family = (father, mother, child1, child2);

var birthday: array[family] of date;

где date – описанная выше запись.

Переменная birthday есть массив, состоящий из записей – дат рождения членов семьи: отца, матери, 1-го ребенка, 2-го ребенка. Каждая дата рождения имеет тип date, который может быть описан в программе.

Для занесения даты рождения, например, mother, достаточно выполнить операторы:

birthday[mother].month := 5;

birthday[mother].day := 1;

birthday[mother].year := 1965;

В то время как массив – объединение компонент одинакового типа, в записи можно объединять компоненты различного типа. Сравните массив и запись:

type

    numb = array[1..4] of integer;

    date = record

        month: 1..12;

        day: 1..31;

        year: integer;

        week: (mo, tu, we, th, fr, sa, su)

    end;

Переменные, определенных выше типов, содержат либо целые массивы, либо целые записи:

var

    m, n: numb;

    one, two: date; 

Присваивание значений элементам массивов и записей также имеет некоторое сходство: 

m[1] := 498;

n[3] := 637;

one.day := 25;

one.week := fr;

two.year := 2011;

Следует отметить, что записи сами могут быть элементами массива. Пример массива записей:

var

    daysopen: array[1..10] of date;

Этот оператор используется для удобства работы с переменными типа record (запись).

Общий вид:

with a do st;

здесь a – имя переменной типа record, st – оператор.

В операторе st при ссылках на компоненты записи имя a можно опускать. 

Пример. Для занесения даты рождения достаточно выполнить операторы:

with birthday[mother] do begin

    month := 5;

    day := 1;

    year := 1965

end;

Общий вид:

type v = record

    a: type1

    b: type2

    ……

    case n: typen of

        c1:(t11: type11;

            t12: type12; …);

        c2:(t21: type21;

            t22: type22; …);

        ……..

end;

var z: v;

Здесь z – переменная типа v; n – переменная, называемая переключателем; typen – тип переменной n.

Этому же типу должны принадлежать метки c1 и c2. Каждой метке соответствует набор полей t11, …. Эти поля являются компонентами варианта.

Если какой-либо метке c вообще не соответствуют поля, то пишут c…: ( );

Замечания:

1. Любая запись (record) может иметь только одну вариантную часть (case).

2. Вариантная часть должна помещаться после постоянной части.

3. Среди идентификаторов полей не должно быть одинаковых. 

Обращение к компоненте z.tij записи z происходит так:

1. Присваивается соответствующее значение c… переключателю n. В зависимости от значения n переменная z, помимо полей a, b, …, содержит те поля, которые соответствуют той метке c…, с какой совпадает значение n.

2. Выполняется операция с компонентой z.tij.
Пусть необходимо собрать следующие сведения о сотрудниках института: фамилию, дату рождения и, если есть семья, то фамилию и дату рождения супруги (супруга).

Эта информация может быть описана, например, записью person.

Пусть переменная типа kind может иметь одно из значений («женат», «холост»).

kind = (married, single);

person = record

    name: string;

    datebirth: date;

    case yesno: kind of 

        married: (name1: string; date1: date);

        single: ()

end;

Здесь name – строка символов; datebirth – запись, содержащая дату рождения; yesno – переключатель типа kind, который может принимать одно из двух значений: married либо single; name1 – строка символов, содержащая фамилию супруга(и); date1 – запись, содержащая дату рождения; single – пустое поле.

Если персона в браке, то будет присутствовать поле married, если нет – поле single. Паскаль допускает вложение вариантов в типе record.

Замечание. Перед засылкой информации в запись программист должен присвоить переключателю соответствующее значение. В противном случае информация в поле заслана не будет, и система никакой диагностики не выдаст.

type
    kind = (married, single);

    date = record
        day: 1..31;

        month: 1..12;

        year: integer;

    end;

    person = record
        name: string;

        datebirth: date;

        case yesno: kind of
            married: (name1: string; date1: date);

            single: ()

    end;

 

var
    p: person;

 

begin
    with p do begin
        yesno := married;

        name := 'panin';

        with datebirth do begin
            day := 14;

            month := 5;

            year := 1962

        end;

        case yesno of
            married: begin
                name1 := 'panina';

                with date1 do begin
                    day := 30;

                    month := 12;

                    year := 1968

                end;

            end;

        end;

    end;

 

    with p do
        write(' ', name);

    with p.datebirth do
        writeln(' ', day, '/', month, '/', year);

    with p do
        write(' ', name1);

    with p.date1 do
        write(' ', day, '/', month, '/', year);

 

readln

end.

Решение задач
Задача1:
Создать каталог из журналов и статей. Выдавать информацию о публикациях, удовлетворяющих тому или иному критерию, например, изданных с 2000 года.

Программа на языке Паскаль: 

type
    t_pub = record
        author: string[40];

        title: string[100];

        year: integer;

        case t: char of
            'b': (publ:string[20]);

            'j': (jour:string[30]; num:byte);

    end;

var
    catalog: array[1..20] of t_pub;

    n,i: integer;

begin
    write('Количество записей: ');

    readln(n);

    for i:=1 to n do begin
        write('Книга или журнал: ');

        readln(catalog[i].t);

        write('Автор: ');

        readln(catalog[i].author);

        write('Название: ');

        readln(catalog[i].title);

        write('Год издания: ');

        readln(catalog[i].year);

        case catalog[i].t of
            'b': begin
            write('Издательство: ');

            readln(catalog[i].publ);

            end;

            'j': begin
            write('Журнал: ');

            readln(catalog[i].jour);

            write('Номер: ');

            readln(catalog[i].num);

            end;

        end;

        writeln;

    end;

 

    for i:=1 to n do
        with catalog[i] do
            if (year >= 2000) and (t = 'b') then
                writeln(author,' ',title,' ',publ,' ',year);

    for i:=1 to n do
        with catalog[i] do
            if (year >= 2000) and (t = 'j') then
                writeln(author,' ',title,' ',jour,' ',year,' ',num);

 

readln;

end.

Задача2: 

Вывести на экран фамилии студентов, чей балл выше среднего по группе.

Описание переменных: 

arr_studs – группа студентов;
average – средний балл в группе.

Программа на языке Паскаль: 

type
    studs = record
        sname: string[25];

        eval: integer;

    end;

var
    arr_studs: array[1..15] of studs;

    sum: integer;

    average: real;

    i,n: byte;

begin
    write('Количество студентов: ');

    readln(n);

    sum := 0;

    for i:=1 to n do
        with arr_studs[i] do begin
            write(i,'-й студент: ');

            readln(sname);

            write('   Оценка: ');

            readln(eval);

            sum := sum + eval;

        end;

    average := sum / n;

writeln;

    writeln('Студенты с оценкой выше средней:');

    for i:=1 to n do
        if arr_studs[i].eval > average then
            writeln(arr_studs[i].sname);

readln

end.

Задача3: 

В текстовом файле содержатся сведения о количестве учащихся в классах и их классных руководителях. Количество классов в школе не превышает 20. Для описания каждого класса отводится одна строка. В первых четырех позициях каждой строки файла записано имя класса (например, 5a, 10b) класса (не пишите русских букв), а затем количество человек в нем и ФИО руководителя. Программа по запросу выдает сведения о классе или сообщение о том, что класс не найден.

Описание переменных: 

school – массив записей, считанных из файла;
title_search – введенное пользователем имя класса. 

Алгоритм решения задачи: 

1. Данные из файла считываются в массив записей. Каждая запись содержит имя класса (title), количество учеников (pupils), сведения о руководителе (teacher). 

2. Пользователь вводит имя класса. Если его длина равна 0, то происходит выход из цикла с помощью команды break.

3. Строка, введенная пользователем, дополняется до 4-х символов пробелами.

4. Если имя класса не совпадет ни с одним полем title записей массива, то флаговая переменная found останется равной false, появится сообщение об отсутствии указанного класса.

5. В случае совпадения все поля соответствующей записи будут выведены на экран.

Программа на языке Паскаль: 

const n = 20;

type
    str4 = string[4];

    clas = record
        title: str4;

        pupils: byte;

        teacher: string[25];

    end;

var
    school: array[1..n] of clas;

    i,j,len: integer;

    title_search: str4;

    found: boolean;

    f: text;

begin
    assign(f,'c:/school.txt'); reset(f);

    i := 1;

    while not Eof(f) do begin
        with school[i] do 

            readln(f,title,pupils,teacher);

        i := i + 1;

    end;

    while true do begin
        write('Input class: ');

        readln(title_search);

        len := length(title_search);

        if len = 0 then break;

        for j := len + 1 to 4 do
            title_search := title_search + ' ';

        found := false;

        for j := 1 to i - 1 do begin
            if title_search <> school[j].title then continue;

            with school[j] do
                writeln(title:5,pupils:5,teacher:25);

            found := true;

            break;

        end;

        if not found then
            writeln('No class!');

    end;

end.

Примечания: 

Для выхода их программы достаточно нажать Enter без ввода данных.

В текстовом файле число, обозначающее количество человек в классе, должно начинаться с 5-й позиции.

Задача4: 

Условие и цель задачи. Пользователь вводит дату в формате ‘dd.mm.yyyy’, которая присваивается строчной переменной s. Необходимо преобразовать строку в запись d, имеющую три числовых поля – day, month, year.

Алгоритм решения задачи: 

Алгоритмы преобразования строчного представления даты в численное могут использоваться в программах, которые выполняют какие-либо вычисления, связанные с датами.

Мы не будем писать сразу наиболее эффективную и короткую программу. Сначала алгоритм решения будет достаточно прозрачным и простым.

Программа на языке Паскаль: 

Алгоритм 1. Линейная программа

Введенная пользователем дата находится в строчной переменной s. 

Извлечение любого символа из строки производится так же, как получение элемента массива по его индексу. Например, s[1] приведет к извлечению первого символа сроки. Можно даже извлечь срез: s[1..2] извлечет первые два символа. Однако как их преобразовать в число?

Функция ord возвращает порядковый номер переданного ей аргумента. Если ей передать символ, то она вернет его номер в таблице символов. Это будет величина целого типа. Например, символ ‘0’ (ноль) может быть 48-м. Поэтому ord(‘0’) вернет 48. Далее номера цифр идут по порядку друг за другом: ‘1’ – 49, ‘2’ – 50, ‘3’ – 51 и т.д.

Чтобы перевести символьное представление цифры (например, ‘7’) в число (7), достаточно выполнить такое выражение: ord(‘7’) – ord(‘0’). Разница номеров между данными символами как раз составляет семь единиц. Таким образом, мы можем перевести символьное представление цифры в число.

Следующая проблема заключается в том, что числа в дате двузначные (день и месяц) и четырехзначные (год). Мы же получили значение только отдельных цифр. Однако, имея цифры и зная их позицию в числе, нетрудно вычислить само число. Например, если первое число 2 (десятки), а второе 4 (единицы), то получить 24 легко: 2 * 10 + 4. Год wxyz вычисляется по формуле w * 1000 + x * 100 + y * 10 + z.

Кроме того, следует не забыть пропустить точки в строке даты. Это символы под номерами 3 и 6. Они никакого участия в вычислениях принимать не должны.

type date = record
        year:1900..2050;

        month:1..12;

        day:1..31;

     end;

 

var
    d: date;

    s: string;

    k,l,m,n: byte;

 

begin
    write('Введите дату в формате dd.mm.yyy: ');

    readln(s);

 

    k := ord(s[1])-ord('0');

    l := ord(s[2])-ord('0');

    d.day := 10 * k + l;

 

    k := ord(s[4])-ord('0');

    l := ord(s[5])-ord('0');

    d.month := 10 * k + l;

 

    k := ord(s[7])-ord('0');

    l := ord(s[8])-ord('0');

    m := ord(s[9])-ord('0');

    n := ord(s[10])-ord('0');

    d.year := 1000 * k + 100 * l + 10 * m + n;

 

    writeln(d.day);

    writeln(d.month);

    writeln(d.year);

 

readln

end.

Алгоритм 2. Использование цикла и массива

Как можно заметить, слишком много раз в предыдущей программе выполняется выражение ord(s[цифра])-ord('0'). Подобные действия лучше выполнять в цикле, а результат помещать не в переменные, а массив.

Требуется преобразовать в цикле все символы строки, за исключением 3-го и 6-го (тех, которые являются точкой), и заполнить массив.

type date = record
        year:1900..2050;

        month:1..12;

        day:1..31;

     end;

 

var
    d: date;

    s: string;

    i, j: byte;

    a: array [1..8] of byte;

 

begin
    write('Введите дату в формате dd.mm.yyy: ');

    readln(s);

 

    i := 1;

    for j := 1 to 10 do
        if s[j] <> '.' then begin
            a[i] := ord(s[j]) - ord('0');

            i := i + 1

        end;

 

    d.day := 10 * a[1] + a[2];

    d.month := 10 * a[3] + a[4];

    d.year := 1000 * a[5] + 100 * a[6] + 10 * a[7] + a[8];

 

    writeln(d.day);

    writeln(d.month);

    writeln(d.year);

 

readln
end.

Алгоритм 3.  Быстрое решение

На самом деле не обязательно использовать массив, цикл и даже лишние переменные. Все делается гораздо проще, однако выражения получаются достаточно длинными и могут быть непонятны для начинающих.

type date = record
        year:1900..2050;

        month:1..12;

        day:1..31;

     end;

 

var
    d: date;

    s: string;

 
begin
    write('Введите дату в формате dd.mm.yyy: ');

    readln(s);

 

    d.day := 10 * (ord(s[1])-ord('0')) + (ord(s[2])-ord('0'));

    d.month := 10 * (ord(s[4])-ord('0')) + (ord(s[5])-ord('0'));

    d.year := 1000 * (ord(s[7])-ord('0')) + 100 * (ord(s[8])-ord('0'));

    d.year := d.year + 10 * (ord(s[9])-ord('0')) + (ord(s[10])-ord('0'));

 

    writeln(d.day);

    writeln(d.month);

    writeln(d.year);

 

readln

end.

Задача5: 

Представим, что нам надо определить принадлежит ли определенная дата диапазону времени, т.е. находится ли между двумя другими датами. 

Две крайние даты пусть задаются программно. А дату, требующую исследования, вводит пользователь. 

Алгоритм решения задачи1: 

Задачу можно решить как минимум двумя способами. Например, для каждой даты вычислить количество дней, прошедших с какой-нибудь начальной даты (например, нулевого года). После чего сравнить полученные значения. Если количество дней анализируемой даты будет находиться между двумя другими, то дата принадлежит диапазону.

Второй, более простой, способ это «поэлементное» сравнение дат: сначала сравниваются года, если этого недостаточно, то месяцы и, затем, дни. Ниже приводится пример реализации этого способа. Однако тут тоже могут быть варианты.

Программа на языке Паскаль: 

type date = record
        year:integer;

        month:1..12;

        day:1..31;

     end;

 

var d1, d2, dmy: date;

 

begin
    with d1 do begin
        year := 1990;

        month := 5;

        day := 15

    end;

    with d2 do begin
        year := 2005;

        month := 12;

        day := 25

    end;

 

    write('Год: ');

    readln(dmy.year);

    write('Месяц: ');

    readln(dmy.month);

    write('День: ');

    readln(dmy.day);

 

    if (dmy.year > d1.year) and (dmy.year < d2.year) then
        writeln('Yes');

 

    if dmy.year = d1.year then
        if dmy.month > d1.month then
            writeln('Yes')

        else
            if dmy.month = d1.month then
                if dmy.day > d1.day then
                    writeln('Yes');

 

    if dmy.year = d2.year then
        if dmy.month < d2.month then
            writeln('Yes')

        else
            if dmy.month = d2.month then
                if dmy.day < d2.day then
                    writeln('Yes');

 

readln

end.

Алгоритм решения задачи2: 

Переменная d имеет тип данных запись и содержит два поля: year – для хранения года и month – для месяца. 

Пользователь вводит год и месяц. 

Оператор выбора case по номеру месяца определяет количество дней в нем. Ситуация осложняется тем, что в феврале может быть как 28 дней, так и 29 (в високосный год).

Год является високосным, если делится без остатка на 4 и при этом не является последним годом столетия, за исключением того столетия, которое без остатка делится на 400.

Программа на языке Паскаль: 

type date = record
        year:1900..2100;

        month:1..12

     end;

 

var d: date;

 

begin
    write('Год: ');

    readln(d.year);

    write('Месяц: ');

    readln(d.month);

 

    case d.month of
        4,6,9,11:

            writeln('В этом месяце 30 дней');

        1,3,5,7,8,10,12:

            writeln('В этом месяце 31 день');

        2:

            if (d.year mod 400 = 0) or
            (d.year mod 100 <> 0) and (d.year mod 4 = 0) then
                writeln('В этом месяце 29 дней')

            else
                writeln('В этом месяце 28 дней');

    end;

 

readln;

end.

Задача 6:   В текстовом файле students.txt хранится информация о студентах: ФИО, год рождения и какую школу окончил. Примеры строк из этого файла

    Иванов Иван Иванович 1985 11
    Петров Петр Петрович 1986 102

Требуется описать массив записей с информацией о студентах и прочитать данные из файла в этот массив. Программа также должна уметь по введенному номеру школы выдавать список закончивших ее студентов.

program Students;

 

type

  {Тип запись с информацией про студента}

  TStudent = record

    F, I, O: string;

    Year: 1900..2100;

    School: 1..200;

  end;

 

var

  {Массив информации о студентах}

  Students: array [1..1000] of TStudent;

  SNum: integer; {Количество студентов}

  SN: 1..200; {Номер школы}

  i: integer;

 

procedure ParseString(s: string; var F, I, O:string;

  var Year, School: integer);

{Процедура, которая из строки, считанной из файла, выделяет

 фамилию, имя, отчество, год рождения и номер школы}

var

  n, code: integer;

begin

  {Положение 1-го пробела в строке}

  n:=Pos(' ', S);

  {Выделение фамилии}

  F:=Copy(S, 1, n-1);

 {Берем все, что идет после 1-го пробела}

  S:=Copy(S, n+1, length(S));

  {Положение 1-го пробела в строке без фамилии}

  n:=Pos(' ', S);

  {Выделение имени}

  I:=Copy(S, 1, n-1);

  {Берем все, что идет после 1-го пробела}

  S:=Copy(S, n+1, length(S));

  n:=Pos(' ', S);

  O:=Copy(S, 1, n-1);

  S:=Copy(S, n+1, length(S));

  n:=Pos(' ', S);

  Val(Copy(S, 1, n-1), Year, code);

  //Year:=StrToInt(Copy(S, 1, n-1)); - для PascaABC

  S:=Copy(S, n+1, length(S));

  n:=Pos(' ', S);

  Val(Copy(S, 1, n-1), School, code);

  //Scool:=StrToInt(Copy(S, 1, n-1));

end;

 procedure ReadData(FileName: string; 

  var Students: array of TStudent; var Num: integer);

{Процедура, читающая данные из файла и определяющая 

 их общее количество}

var

  F: text;

  S: string;

begin

  AssignFile(F, FileName);

  Reset(F);

  Num:=0;

  while not EOF(F) do

  begin

    Readln(F, S);

    with Students[Num] do

      ParseString(S, F, I, O, Year, School);

    Num:=Num+1;

  end;

  CloseFile(F);

end;

 

begin

  ReadData('students.txt', Students, SNum);

  Write(‘Input school number’);

  Readln(sn);

  {Поиск студентов из школы sn}

  for i:=1 to SNum do

    with Students[i] do

    begin

      if School = sn then

        writeln(F, ' ', I, ' ', O);

    end;
  readln;

end.

Решение задач типа С4 из ЕГЭ с использованием записей.
Задача1
На вход в программе подаются сведения о студентах с 1-го по 5-й курс некоторого вуза. В первой строке сообщается количество студентов N, которое больше 20, но меньше или равно 100, каждая из следующих N строк имеет следующий формат: <фамилия> <имя> <курс> <стипендия>, где <фамилия> - строка, состоящая не более чем из 20 символов, <имя> - строка, состоящая не более чем из 15 символов, <курс> - целое число от 1 до 5, <стипендия> - вещественное число. <фамилия>  и <имя> , а также <имя>  и  <курс> и <стипендия> разделены одним пробелом. Входные данные не упорядочены никаким способом.

Пример входных строк: 

25 

Федорова Ирина 5 4500 

Иванов Сергей 3 2050 

Требуется написать программу, которая будет выводить на экран фамилии и имена студентов, имеющих максимальную стипендию на своём курсе в порядке с первого курса по пятый. Если есть студенты с одинаковыми стипендиями, то следует вывести их фамилии и имена. Алфавитный порядок при выводе фамилий можно не соблюдать. Считать, что на всех курсах есть студенты, получающие стипендию.

Пример выходных строк: 

Курс 1 

Петров Иван 

Катаев Сергей 

Курс 2 

Смирнов Максим
Программа :

var p:array[1..100] of record 
name: string;
kurs:integer;
stip: real; 
end; 
c:char; 
i,k,n,m:integer; 
max_kurs:array[1..5] of real; 
begin 
readln(n); 
for i:=1 to n do 
max_kurs[i]:=0;{обнулено мах значение стипендии по всем 5 курсам}
for i:=1 to n do 
begin 
p[i].name:=''; 
repeat 
read(c); 
p[i].name:=p[i].name+c 
until c='  ';{считана фамилия}
repeat 
read(c); 
p[i].name:=p[i].name+c 
until c='  ';{считано имя}
read(m);{считан курс} 
p[i].kurs:=m; 
read(p[i].stip);{считана стипендия} 
if p[i].stip>max_kurs[p[i].kurs] then max_kurs[p[i].kurs]:=p[i].stip; 
readln;
end; {цикла}
{вывод студентов с мах стипендией на курсе}
writeln('максимальная стипендия студентов’);writeln;
for k:=1 to 5 do begin 
writeln('курс',k); 
for i:=1 to n do 
with p[i] do 
begin 
if ((kurs=k)and (stip=max_kurs[k])) then writeln(name); 
end; 
writeln; 
end; 
readln; 
end. 
Задача 2: 

В командных олимпиадах по программированию для решения предлагается

не больше 11 задач. Команда может решать предложенные задачи в любом

порядке. Подготовленные решения команда посылает в единую

проверяющую систему соревнований. Вам предлагается написать

эффективную, в том числе по используемой памяти, программу, которая

будет статистически обрабатывать пришедшие запросы, чтобы определить

наиболее популярные задачи. Следует учитывать, что количество запросов в списке может быть очень велико, так как многие соревнования проходят с

использованием Интернет.

Перед текстом программы кратко опишите используемый вами алгоритм

решения задачи.


На вход программе в первой строке подаётся количество пришедших

запросов N. В каждой из последующих N строк записано название задачи

в виде текстовой строки. Длина строки не превосходит 100 символов,

название может содержать буквы, цифры, пробелы и знаки препинания.

Пример входных данных: 

6 

А+B 

Крестики-Нолики 

Прямоугольник 

Простой делитель 

А+В 

Простой делитель 

Программа должна вывести список из трёх наиболее популярных задач

с указанием количества запросов по ним. Если в запросах упоминаются

менее трех задач, то выведите информацию об имеющихся задачах. Если

несколько задач имеют ту же частоту встречаемости, что и третья по частоте

встречаемости задача, их тоже нужно вывести.

Пример выходных данных для приведённого выше примера входных данных: 

А+В 2 

Простой делитель 2 

Крестики-Нолики 1 

Прямоугольник 1
program C4; 
type avto= record 
d:integer;{d-kolichestvo poiavlenii zadach s1} 
s1:string;{nazvanie zadachi} 
end; 
var 
s:string; 
a : array [1..11] of avto; {massiv sodergahii zapisi} 
n,i,j,num,max,maxi:integer; 
begin 
num:=0;{num - kolichestvo razlichnix zadach} 
readln(n);{n - kolichestvo vsex nazvanii zadach} 
for i:=1 to n do 
begin 
readln(s);{vvodim nazvanie zadachi} 
j:=1; 
while (j<=num) and (s<>a[j].s1) do j:=j+1;{proveriaem bila li do etigo dannoe nazvanie zadachi} 
if j<=num then{esli da} 
a[j].d:=a[j].d+1{pribavliaem odin k schetchiky poiavlenii} 
else begin  {inache} 
     a[j].s1:=s;{zanosim nazvanie v massiv} 
     a[j].d:=1; 
     num:=num+1; 
     end; 
end; 
for i:=1 to num do {ihem maximalnii} 
if max<a[i].d then begin max:=a[i].d; maxi:=i; end; 
writeln('',a[maxi].s1,' ',a[maxi].d); {vivodim ego} 
a[maxi].d:=0;{ochishaem scetchik poiavlenii} 
max:=0; 
if num>1 then    {proveriaem kolichestvo razlichnix zadach } 
begin  {ecli da to snowa vivodim maxsimalnii} 
for i:=1 to num do 
if max<a[i].d then begin max:=a[i].d; maxi:=i; end; 
writeln('',a[maxi].s1,' ',a[maxi].d); 
a[maxi].d:=0; 
max:=0; 
end; 
if num>2 then    {proveriaem kolichestvo razlichnix zadach } 
begin   {ecli da to snowa vivodim maxsimalnii} 
for i:=1 to num do 
if max<a[i].d then begin max:=a[i].d; maxi:=i; end; 
for i:=1 to num do {ecli takix znach mnogo to vivodim vse} 
if max=a[i].d then 
writeln('',a[i].s1,' ',a[i].d); 
end;
readln;
end.
Заключение
Тип-запись включает ряд компонент, называемых полями, которые могут быть разных типов. При задании типа-записи после зарезервированного слова record следует перечислить все поля типа-записи с указанием через двоеточие их типов и завершить задание типа словом end. Поля отделяются друг от друга точкой с запятой. Количество полей записи может быть любым. Если несколько полей типа-записи имеют один и тот же тип, то их имена можно перечислить через запятую и затем указать этот общий тип. После введения типа-записи можно затем задать переменные или типизированные константы этого типа - записи. При задании значений константе ее поля записываются в круглых скобках через точку с запятой. Для задания значения каждого поля сначала указывается имя поля, а затем через двоеточие - его значение. 

Тип-запись можно вводить и непосредственно при определении переменных или типизированных констант. 

Для того чтобы не выписывать каждый раз имя записи при обращении к ее полям, можно использовать оператор над записями WITH. Его структура выглядит следующим образом: with <имя записи> do <оператор>. В  этом  случае внутри  оператора  можно  указывать только  поле  записи. 


Тип-запись может иметь вариантную часть, изменяющуюся при разных реализациях, например, в типе-записи параметров геометрических фигур для квадрата задается сторона, для треугольника - две стороны и угол между ними, для окружности - радиус и т. д. Вариантная часть может быть только одна и должна располагаться в конце записи.


Вариантная часть начинается словом case, за которым следует переменная выбора варианта с указанием типа. Далее указываются константы значения которых может принимать переменная выбора варианта. За каждой константой через двоеточие в круглых скобках записываются поля данного варианта записи с указанием их типов. Круглые скобки необходимы, даже если для данного варианта вариантная часть отсутствует. Следует отметить, что вариантная часть не завершается своей операторной скобкой end, как у обычной конструкции CASE, т. к. далее идет завершающая операторная скобка end всего типа.


В стандарте языка Паскаль перед тем, как использовать один из вариантов записи, переменной выбора варианта необходимо присвоить соответствующее значение:

Fig := Square; 

MySquare.Side :=5.0; 

Fig := Circle; 

MyCircle.Radius :=7.5;


В Turbo Pascal эту операцию делать необязательно, хотя, если переменная выбора задана, ее затем можно в программе посмотреть, чтобы определить, какой задан вариант. Более того, задание значения переменной выбора ни на что не оказывает влияния. В тип-записи допустимо не вводить переменную выбора вариантов, а перечислить лишь разные варианты.

